ActionScript in Flash 

Animated Game 

Summary and Overview 

As you know, Movie Clips are objects within Flash, each with their own independent timeline. As well as being able to pass variables between different Movie Clips, Flash contains methods associated with the Movie Clip object, increasing the ability to manipulate them and other objects within Flash. 

One of the more significant methods associated with the Movie Clip object is the attachMovie method. This method allows Flash to dynamically attach Movie Clips to one another. 

In this lesson you will learn: 

1. To create a simple ‘shoot em up’ style arcade game 

2. To use Movie Clip methods, including attachMovie, hitTest and removeMovieclip 

3. To use new syntax in ActionScript as well as the return and eval functions and for loops 

Exercise 1a: 

Create a movie with a rifle that follows the mouse position horizontally and shoots bullets at a shelf of bottles. Display ammo statistics that decrease when fired. A shot sound will be played on mouse up. The bottles will break on collision with the bullet, play a smashing sound and display a message. Once all bullets have been used, or all bottles have been broken, display a ‘play again?” button to reset the game. Use bottles.fla for graphical content. 

Looking at the finished product 

1. Open Macromedia Flash MX 

2. Open bottlesFinished.swf 

3. Test the completed movie 

4. When you have finished viewing the file, close bottlesFinished.swf 

Setting up the Movie 

	Open the bottles.fla file and then open the library within it. There should be several graphics and a range of Movie Clips. Let’s start by tidying up the library so we can easily find elements. 

Using the ‘New Folder’ option from the menu at the top right of the Library title bar, create a new folder called ‘GFX’ (the common abbreviation for graphics). Hold down shift and select the graphic elements within the library and drag them into the GFX folder. 

	Create three other folders: BTN, MC and AUDIO. Sort the remaining library content into the appropriate folder. 


	Let’s set up our timeline ready to receive our content. Select frame 5 in the actions layer and drag downwards to select frame five in the other layers beneath. Right click and select ‘Insert Blank Keyframe’ 

This will insert a blank keyframe in every layer which you have selected. A real timesaver when you are using so many layers! 


Setting up the timeline 

Starting in the ‘shelf’ layer, with the blank keyframe in frame five selected, open the Library and drag an instance of ‘gfx_shelf’ to the stage. Position it about two thirds of the way up the stage and in the centre horizontally. Extend the endframe of the object to frame 25 in the timeline. You can do this by holding down Ctrl+Alt and dragging. 

On the next layer up, with the blank keyframe in frame 5 selected, drag eight bottle Movie Clips to the stage. Position them so they appear as though they are sitting on the shelf. It doesn’t matter which order you position them in, but you will need to make sure the instance names are bottle01, bottle02, bottle03 etc across the stage, left to right. There are already shadows on the shelf to help you position the beer bottles, and you can vertically align their bottoms by selecting them all and pressing Ctrl+Alt+6. When you’re done positioning the bottles, hold down Ctrl+Alt and drag the endframe to frame 25. 

In the ‘playAgain’ layer, insert a blank keyframe in frame 25 (or drag the one already in frame 5 there) and position an instance of btn_playAgain in the centre of the stage just below the shelf of bottles. At frame 25 your stage should now look like this. 

Starting in the ‘shelf’ layer, with the blank keyframe in frame five selected, open the Library and drag an instance of ‘gfx_shelf’ to the stage. Position it about two thirds of the way up the stage and in the centre horizontally. Extend the endframe of the object to frame 25 in the timeline. You can do this by holding down Ctrl+Alt and dragging. 

On the next layer up, with the blank keyframe in frame 5 selected, drag eight bottle Movie Clips to the stage. Position them so they appear as though they are sitting on the shelf. It doesn’t matter which order you position them in, but you will need to make sure the instance names are bottle01, bottle02, bottle03 etc across the stage, left to right. There are already shadows on the shelf to help you position the beer bottles, and you can vertically align their bottoms by selecting them all and pressing Ctrl+Alt+6. When you’re done positioning the bottles, hold down Ctrl+Alt and drag the endframe to frame 25. 

In the ‘playAgain’ layer, insert a blank keyframe in frame 25 (or drag the one already in frame 5 there) and position an instance of btn_playAgain in the centre of the stage just below the shelf of bottles. At frame 25 your stage should now look like this. 

Continuing on to the gun layer, with frame 5 selected add and instance of the mc_gun to the stage. Position it in the horizontal centre of the stage with the bottom just below that of the stage. Give the Movie Clip an instance name of ‘gun’. Extend the endframe out to frame 25. 

On the ‘control’ layer in frame 5 select the Movie Clip mc_controller in the library and drag it to a position off the stage. As per last week’s exercise, what the controller looks like is unimportant: it is simply used to hold ActionScript instructions. While you’re on this layer, add some static text ‘Ammo’ and place beneath it a dynamic text field with the variable name of ‘ammo’. Select both these elements and convert them to a Movie Clip called ‘mc_displayAmmo’. Extend the endframe to frame 14. 

Finally on the ‘actions’ layer, add blank keyframes in frames 15 and 25. Select the blank keyframe in frame 5 and give it the frame label ‘game’. In frame 15 give the keyframe the frame label ‘finish’. If all has gone according to plan, your file should now look something like this. 

Before we go on to add the ActionScript to get everything working, don’t forget to clean up your library by ‘filing’ the extra elements we’ve just created. You can expand or contract the folders in the library by double clicking on the icons beside the folder name. 

Adding ActionScript to make it work 
Click in the keyframe in frame 1 of the actions layer and type the following code: 

fscommand("allowscale",false); 

maximumBullets = 10; 

numberOfBulletsFired = 0; 

totalNumberOfTargets = 8; 

gameOver = false; 

myTargetsState = new Array(); 

myTargetsState = [0,0,0,0,0,0,0,0]; 

myTargets = new Array(); 

myTargets = ["bottle01","bottle02","bottle03","bottle04","bottle05","bottle06","bottle07","bottle08"]; 

function resetGame() 

{ 

_root.numberOfBulletsFired = 0; 

_root.gameOver = false; 

_root.myTargetsState = new Array(); 

_root.myTargetsState = [0,0,0,0,0,0,0,0]; 

_root.myTargets = new Array(); 

_root.myTargets = ["bottle01","bottle02","bottle03","bottle04","bottle05","bottle06","bottle07","bottle08"]; 

_root.resetBottles(); 

_root.gotoAndStop("game"); 

} 

function resetBottles() 

{ 

for(x=0;x < _root.myTargets.length;x++) 

{ 

temp = eval(_root.myTargets[x]); 

temp.gotoAndStop("norm"); 

temp.setMode(); 

} 

} 

function fireBullet() 

{ 

if (_root.gameOver == false) 

{ 

if (_root.numberOfBulletsFired < _root.maximumBullets) 

{ 

myGunShot = new sound(); 

myGunShot.attachSound("shortshot"); 

myGunShot .start(); 

_root.numberOfBulletsFired++; 

this.attachMovie("bullet","bullet" + numberOfBulletsFired,numberOfBulletsFired * 10); 

this["bullet" + numberOfBulletsFired]._x = _root.gun._x; 

this["bullet" + numberOfBulletsFired]._y = _root.gun._y; 

} 

} 

} 

function checkForHit(whichBullet) 

{ 

for(x=0;x < _root.myTargets.length;x++) 

{ 

temp = eval(_root.myTargets[x]); 

if (whichBullet.hitTest(temp)) 

{ 

if (_root.myTargetsState[x] == 0) 

{ 

if (temp._visible == 1) 

{ 

_root.destroyBullet(whichBullet); 

temp.gotoAndPlay("hit"); 

temp.setActiveState(0); 

_root.myTargetsState[x] = 1; 

} 

} 

} 

} 

} 

function destroyBullet(whichBullet) 

{ 

removeMovieClip(whichBullet); 

} 

function checkForFinish() 

{ 

hitTotal = 0; 

for(x=0;x < _root.myTargetsState.length;x++) 

{ 

hitTotal = hitTotal + _root.myTargetsState[x]; 

} 

if (hitTotal == _root.totalNumberOfTargets) 

{ 

_root.gameOver = true; 

_root.gotoAndPlay("finish"); 

return 

} 

if (_root.numberOfBulletsFired == _root.maximumBullets) 

{ 

_root.gameOver = true; 

_root.gotoAndPlay("finish"); 

return 

} 

} 

While it looks like a lot of code, it’s really not that complex. Let’s go through it block by block and see what each part does. The first section: 

fscommand("allowscale",false); 

maximumBullets = 10; 

numberOfBulletsFired = 0; 

totalNumberOfTargets = 8; 

gameOver = false; 

is similar to what we have done before. We’re setting up initial states of variables which we will use later in our functions as well as making a command relating to how the movie will be displayed. The line fscommand("allowscale",false)uses the built in Flash fscommand. This command lets the Flash movie communicate with either the Flash Player, or the program hosting the Flash Player, such as a web browser. In this instance it is telling the host not to allow the movie to be scaled. 

The next block: 

myTargetsState = new Array(); 

myTargetsState = [0,0,0,0,0,0,0,0]; 

myTargets = new Array(); 

myTargets = ["bottle01","bottle02","bottle03","bottle04","bottle05","bottle06","bottle07","bottle08"]; 

sets up two arrays, one called myTargetsState, the other called myTargets. MyTargetsState holds 8 blank values of zero (signified by the 8 comma separated values). The second array, myTargets, also holds 8 values: bottle01, bottle02, bottle03, bottle04, bottle05, bottle06, bottle07 and bottle08. The inverted commas around each value indicates this array holds string values. 

The next block of code is setting up the function resetGame. Remember, this is not actually activating this code: it’s setting it up so that it can be called from elsewhere within the program. 

function resetGame() 

{ 

_root.numberOfBulletsFired = 0; 

_root.gameOver = false; 

_root.myTargetsState = new Array(); 

_root.myTargetsState = [0,0,0,0,0,0,0,0]; 

_root.myTargets = new Array(); 

_root.myTargets = ["bottle01","bottle02","bottle03","bottle04","bottle05","bottle06","bottle07","bottle08"]; 

_root.resetBottles(); 

_root.gotoAndStop("game"); 

} 

When this code is executed, Flash sets the value of the variable numberOfBulletsFired to zero. Similarly it sets the value of the Boolean variable gameOver to false. The next four lines deal with the arrays we were just discussing. Essentially it resets the values within each array to their initial values. The line _root.resetBottles() executes the resetBottles function and finally root.gotoAndStop("game") tells Flash to gotoAndStop at the frame marker game in the root timeline. Basically the resetGame function returns everything to its initial state so another game can be played. 

The next block contains the function resetBottles. 

function resetBottles() 

{ 

for(x=0;x < _root.myTargets.length;x++) 

{ 

temp = eval(_root.myTargets[x]); 

temp.gotoAndStop("norm"); 

temp.setMode(); 

} 

} 

This function contains a for loop. For loops are a bit like while loops. They contain three parts; a init (initialize) expression, a condition and a next expression. Essentially the init expression is evaluated once, the condition is evaluated as either a true or false and provided the evaluation proves true the code within the curly brackets is executed and the next expression is executed after each loop iteration. It’s important to remember that the init expression is only read the once. 

What the above code does is set the variable x to zero, checks that the value of variable x is less than that of myTargets.length, executes the code within the brackets and then increments the variable x by one before re-evaluating the condition. We know that myTargets is an array (with a length of 8 values) so the for loop above will execute the code while x is less than 8. In other words, eight times. The inner code tells Flash to look inside (or evaluate) position x in the array myTargets and assign its value to the variable temp (as you will remember, the array contains instance names for our bottles). 

The second line tells Flash to gotoAndStop at the frame marker ‘norm’ in the Movie Clip whose instance name is held within the variable ‘temp’. Once this is done, the for statement will add one to the variable x and then check to see if x is still smaller than the number of positions in the array. If it is, the code within the brackets will be executed again but this time (because x is one value higher) temp will be assigned the value of the next position in the array. 

The next block of code contains the fireBullet function. 

function fireBullet() 

{ 

if (_root.gameOver == false) 

{ 

if (_root.numberOfBulletsFired < _root.maximumBullets) 

{ 

myGunShot = new sound(); 

myGunShot.attachSound("shortshot"); 

myGunShot .start(); 

_root.numberOfBulletsFired++; 

this.attachMovie("bullet","bullet" + numberOfBulletsFired,numberOfBulletsFired * 10); 

this["bullet" + numberOfBulletsFired]._x = _root.gun._x; 

this["bullet" + numberOfBulletsFired]._y = _root.gun._y; 

} 

} 

} 

This block contains an if statement within an if statement! Sounds complicated, but it’s much easier than it initially looks! The first if statement if (_root.gameOver == false) uses the equality operator to check if the variable gameOver in the root timeline is equal to false. If it is, the code within the brackets (the second if statement) will be executed. 

The second if statement checks if the value in the variable numberOfBulletsFired within the root timeline is less than the value of the maximumBullets variable. If it is, the code within the inner brackets is executed. 

The inner code: 

myGunShot = new sound(); 

myGunShot.attachSound("shortshot"); 

myGunShot .start(); 

_root.numberOfBulletsFired++; 

this.attachMovie("bullet","bullet" + numberOfBulletsFired,numberOfBulletsFired * 10); 

this["bullet" + numberOfBulletsFired]._x = _root.gun._x; 

this["bullet" + numberOfBulletsFired]._y = _root.gun._y; 

contains a new sound object myGunShot as well as incrementing the variable numberOfBulletsFired by one. The second block uses the attachMovie method of the Movie Clip object to attach a Movie Clip to the current Movie Clip (the one from which the fireBullet function is called). 

The attachMovie method is made up of four parts: the Movie Clip name, the id name, the newname and the depth. The Movie Clip name is the name of the Movie Clip to which the new Movie Clip is being attached, the id name is the name of the Movie Clip in the library to attach, the newname sets a unique instance name for the Movie Clip being attached and the depth is an integer specifying at which level the movie should be placed. 

In the case of the code above, the ‘this’ keyword is used to tell Flash that the Movie Clip to attach the movie to is whichever Movie Clip the function is called from (in our case the mc_gun Movie Clip). The Movie Clip ‘bullet’ is the one to be attached and it is to be given the instance name of ‘bullet’ + the value of the variable numberOfBulletsFired (e.g. bullet5). The numberOfBulletsFired variable is again used; this time it is multiplied by ten to give the level at which the Movie Clip should be placed. 

A problem? Yes indeed! There IS no bullet Movie Clip, is there? Do you remember how Flash refers to unloaded objects in its library? Through the name specified in the Linkage Properties panel. To create a Movie Clip referred to as ‘bullet’, click on the mc_bronzeBullet in the library, drag an instance to any blank position on the main timeline and convert it into a Movie Clip called ‘mc_bullet’. Delete the new Movie Clip from the timeline, select mc_bullet in the library, right click and select ‘Linkage’. In the Linkage Properties panel, name the Movie Clip ‘bullet’ and make sure the ‘Export for ActionScript’ and ‘Export in First Frame’ options are selected. 

The last two lines within this block: 

this["bullet" + numberOfBulletsFired]._x = _root.gun._x; 

this["bullet" + numberOfBulletsFired]._y = _root.gun._y; 

tell Flash to position the instance ‘bullet’ + numberOfBulletsFired (the instance of the Movie Clip we have just attached) to the same horizontal and vertical position as the instance of gun in the root timeline. 

The next function checks to see if the bullet has hit a target. 

function checkForHit(whichBullet) 

{ 

for(x=0;x < _root.myTargets.length;x++) 

{ 

temp = eval(_root.myTargets[x]); 

if (whichBullet.hitTest(temp)) 

{ 

if (_root.myTargetsState[x] == 0) 

{ 

if (temp._visible == 1) 

{ 

_root.destroyBullet(whichBullet); 

temp.gotoAndPlay("hit"); 

temp.setActiveState(0); 

_root.myTargetsState[x] = 1; 

} 

} 

} 

} 

} 

The checkForHit function uses the local variable whichBullet to pass a value into the function. Don’t worry about what value whichBullet holds – we’ll get to that later when we add the code to our bullet. The for loop is the same as the one we encountered earlier; set the variable x to zero, check that the value 

of variable x is less than that of myTargets.length, execute the code within the brackets and then increment the variable x by one before re-evaluating the condition. 

The statement within the brackets 

temp = eval(_root.myTargets[x]); 

if (whichBullet.hitTest(temp)) 

{ 

if (_root.myTargetsState[x] == 0) 

{ 

if (temp._visible == 1) 

{ 

_root.destroyBullet(whichBullet); 

temp.gotoAndPlay("hit"); 

temp.setActiveState(0); 

_root.myTargetsState[x] = 1; 

} 

} 

} 

uses the eval function to look inside (or evaluate) position x in the array myTargets and assign its value to the variable temp (as you will remember, the array contains instance names for our bottles). It then uses the Movie Clip method hitTest to determine whether the bullet whichBullet has entered the area on the stage occupied by the object within the variable temp. Put simply, whether the bullet has hit the specific bottle. 

If the bullet has occupied the same area as the bottle, then Flash is told to destroy the bullet (calling the destroyBullet function and passing to it the value whichBullet), to gotoAndPlay the ‘hit’ frame marker in the instance of the Movie Clip being held within the temp value and to set the value of the position in the array myTargetsState designated by the x to 1. 

The destroyBullet function we just mentioned is really simple; it uses removeMovieClip to tell Flash to remove the Movie Clip whichBullet. 

function destroyBullet(whichBullet) 

{ 

removeMovieClip(whichBullet); 

} 

The last block on this frame checks whether or not the game is finished. By now most of the syntax here should be familiar to you. In fact the only new thing you’ll see used is the return function. The return function specifies the value returned by a function. The return action evaluates the expression and returns the result as a value of the function in which it executes. The return action causes the function to stop running and replaces the function with the returned value. If the return statement is used alone, it returns null. 

function checkForFinish() 

{ 

hitTotal = 0; 

for(x=0;x < _root.myTargetsState.length;x++) 

{ 

hitTotal = hitTotal + _root.myTargetsState[x]; 

} 

if (hitTotal == _root.totalNumberOfTargets) 

{ 

_root.gameOver = true; 

_root.gotoAndPlay("finish"); 

return 

} 

if (_root.numberOfBulletsFired == _root.maximumBullets) 

{ 

_root.gameOver = true; 

_root.gotoAndPlay("finish"); 

return 

} 

} 

Phew! That’s it for frame 1! When you’re done, add stop actions to frames 5 and 25. Thankfully there’s not much else in the way of coding in this file: a small amount on the gun instance on the stage, another little bit on the mc_bronzeBullet and on the play again button. Select the gun instance on the stage, open the Actions panel and type the following code: 

onClipEvent (enterFrame) 

{ 

_x = _root._xmouse; 

} 

onClipEvent (mouseUp) 

{ 

if (_root.gameOver == false) 

{ 

_root.fireBullet(); 

} 

} 

The first block instructs Flash to make the horizontal position of the gun follow that of the mouse. The second statement tells Flash that on mouse up, if the variable gameOver is equal to false, execute the function fireBullet. 

Open the library and double click the icon beside the mc_bullet Movie Clip. Select the mc_bronzeBullet on the stage and add the following ActionScript: 

onClipEvent (load) 

{ 

pSpeed = 20; 

} 

onClipEvent (enterFrame) 

{ 

_parent._y = _parent._y - pSpeed; 

if (_parent._y < 0) 

{ 

removeMovieClip(_parent); 

} 

_root.checkForHit(_parent); 

_root.checkForFinish(); 

} 

You’ll notice where we call the function checkForHit the value being passed to its _parent – e.g. the parent Movie Clip of the bullet. That’s where the whichBullet value we used earlier comes from! 

Now we need to add some ActionScript to the mc_displayAmmo instance on the stage. Click on it and add the following code: 

onClipEvent (enterFrame) 

{ 

ammo = (_root.maximumBullets - _root.numberOfBulletsFired); 

} 

We also need to add a piece of code to the controller Movie Clip to see whether the game has ended. Click on the instance of mc_controller and type in this code: 

onClipEvent (enterFrame) 

{ 

_root.checkForFinish(); 

} 

Finally add the following code to the btn_playAgain in frame 25: 

on (release) 

{ 

_root.resetGame(); 

} 

Save your movie as ‘bottlesFinished.fla’ and test. 

Exercise 1b: 

You’re finished exercise 1a and have a working arcade style game. But it’s pretty boring. There’s not much of a challenge shooting at bottles that stay still from such close range, is there?! To liven things up a bit (and challenge your ActionScript skills!) we’re now going to make a few revisions to our previous file. 

In this second part of the exercise we are going to make the game a little more challenging by making the bottles on the shelf move and by adding a timer. 

Looking at the finished product 

1. Open bottlesFinished1.swf 

2. Test the completed movie 

3. When you have finished viewing the file, close bottlesFinished1.swf 

Setting up the file 

Use the bottlesFinished.fla movie you created earlier, but before you change it, save your new file as bottlesFinished1.fla. An important thing to remember is NEVER overwrite your original file – you never know when you may need to revert to it! 

The first thing you need to do is to create another layer on the main timeline. Create a new layer above the ‘control’ layer and call it ‘timer’. Insert blank keyframe in frame 5. In the blank keyframe in frame 5, create static text ‘Time Left’ and position it near the ‘Ammo’ text you have on the stage already. Under the new heading, create a dynamic text field with the variable name ‘gameTimer’. Select the static and dynamic text and covert them to a Movie Clip called ‘mc_timer’. 

Click on the instance of mc_timer on the stage and type the following code: 

onClipEvent (load) 

{ 

startGameTime = getTimer()/1000; 

duration = 30; 

this.gameTimer = "30"; 

countDown = 0; 

} 

onClipEvent (enterFrame) 

{ 

if (int(this.gameTimer) > 0) 

{ 

timePassed = ((getTimer()/1000) - this.startGameTime); 

this.gameTimer = int(duration - timePassed); 

} 

else 

{ 

_root.gotoAndPlay("finish"); 

} 

} 

The first section of code uses the getTimer function in Flash to determine the number of milliseconds which have elapsed since the Movie Clip started playing, divide that value by 1000 and assign the result to the variable startGameTime. Initial and constant values are also set up here, such as the nominated duration, the initial value of the variable gameTimer within the current Movie Clip and a value of zero for countDown. 

The second section uses an if statement to determine whether the integer value of the variable gameTimer within the current Movie Clip is greater than zero. If it is, it executes the code: 

timePassed = ((getTimer()/1000) - this.startGameTime); 

this.gameTimer = int(duration - timePassed); 

which divides the value within the getTimer variable by 1000 and subtracts from this the startGameTime value before assigning the result to the timePassed variable. It then sets the value of the gameTimer variable within the current Movie Clip to be the integer of duration minus timePassed. 

If the integer value of gameTimer is not greater than zero, Flash is told to gotoAndPlay the frame label ‘finish’ in the root timeline. 

The next section of ActionScript to add is quite long. It’s the actions we’re putting on the bottles to make them move. Lucky for us, the code is the same for each of the bottles, so we can write it once and then add it to our bottles! 

Select the first bottle instance on your stage and add the following ActionScript: 

onClipEvent (load) 

{ 

movementArray = new Array(); 

movementArray = ["move","visible"]; 

stageWidth = 620; 

pActive = 1; 

pOriginalX = _x; 

this.setMode(); 

pDuration = 2; 

function resetTimer() 

{ 

startTime = getTimer()/1000; 

} 

function setActiveState(whichState) 

{ 

this.pActive = whichState; 

} 

function setMode() 

{ 

_visible = 1; 

_x = this.pOriginalX; 

pMode = movementArray[random(2)]; 

pMovement = random(20) + 5; 

pActive = true; 

this.restTimer(); 

} 

} 

onClipEvent (enterFrame) 

{ 

if (_root.gameOver == false) 

{ 

if (pActive) 

{ 

if (pMode == "move") 

{ 

if (_x > stageWidth) 

{ 

_x = pOriginalX; 

} 

x = _x + pMovement; 

} 

if (pMode == "visible") 

{ 

if (getTimer()/1000 > (startTime + this.pDuration)) 

{ 

if (_visible == 1) 

{ 

_visible = 0; 

this.resetTimer(); 

} 

else 

{ 

_visible = 1; 

this.resetTimer(); 

} 

} 

} 

} 

} 

} 

If you read through the code, most of it should be pretty familiar to you now! The first block sets the initial values of variables and constants. It then sets up the functions we call later in the program. The random function is new to you – it specifies a random value between 0 and the value specified (in our case 20). 

The second block of ActionScript uses a series of if/else statements to execute the code. Again, all of this should be familiar to you now, so read through it and see if you can trace what’s happening. 

Once you’re finished typing the code, copy and paste it onto each bottle instance on the stage. Save and test your file. 

